

Four Problems with Policy-Based Constraints and How to Fix Them

Dillan Mills
Synopsys, Inc.
Maricopa, AZ
dillan@synopsys.com

Chip Haldane
The Chip Abides, LLC
Gilbert, AZ
chip@thechipabides.com

Abstract

This paper presents solutions to problems encountered in the implementation of policy classes for SystemVerilog constraint layering. Policy classes provide portable and reusable constraints that can be mixed and matched into the object being randomized. There have been many papers and presentations on policy classes since the original presentation by John Dickol at DVCon 2015. The paper addresses three problems shared by all public policy class implementations and presents a solution to a fourth problem. The proposed solutions introduce policy class inheritance, tightly pair policy definitions with the class they constrain, reduce the expense of defining common policies using macros, and demonstrate how to treat policies as disposable and lightweight objects. The paper concludes that the proposed solution improves the usability and efficiency of policy classes for SystemVerilog constraint layering.

I. CONSTRAINTS AND POLICY CLASS REVIEW

Random objects and constraints are the foundational building blocks of constrained random verification in SystemVerilog. The simplest implementations embed fixed constraints within a class definition. Embedded constraints lack flexibility; all randomized object instances must meet these requirements exactly as they are written.

In-line constraints using the **with** construct offer marginally better flexibility. Although these external constraints allow greater variability of random objects, their definitions are still fixed within the calling context. Furthermore, all in-line constraints must be specified within a single call to **randomize()**.

Policy classes are a technique for applying SystemVerilog constraints in a portable, reusable, and incremental manner, originally described by John Dickol [1] [2]. The operating mechanism leverages an aspect of "global constraints," the simultaneous solving of constraints across a set of random objects. Randomizing an object that contains policies also randomizes the policies. Meanwhile, the policies contain a reference back to the container. Consequently, the policy container is constrained by the policies it contains. Dickol's approach is illustrated by the following code.

```
1  class policy_base#(type ITEM=uvm_object);
2      ITEM item;
3
4      virtual function void set_item(ITEM item);
5          this.item = item;
6      endfunction
7  endclass
8
9  class policy_list#(type ITEM=uvm_object) extends policy_base#(ITEM);
10     rand policy_base#(ITEM) policy[$];
11
12     function void add(policy_base#(ITEM) pcy);
13         policy.push_back(pcy);
14     endfunction
15
16     function void set_item(ITEM item);
17         foreach(policy[i]) policy[i].set_item(item);
18     endfunction
19 endclass
```

Fig. 1: The **policy_base** and **policy_list** classes

The first two classes provide the core definitions for policies. The **policy_base** implements the hook back to the policy container, and the **policy_list** enables related policies to be organized into groups. Both of these classes are parameterized by a container object type, so a unique specialization will be required for each policy container class that needs to be constrained.

Next, **addr_txn** is an example policy container—a generic transaction with a random address and size that implements policies.

```

1  class addr_txn;
2      rand bit [31:0] addr;
3      rand int      size;
4      rand policy_base#(addr_txn) policy[$];
5
6      constraint c_size {size inside {1, 2, 4};}
7
8      function void pre_randomize;
9          foreach(policy[i]) policy[i].set_item(this);
10     endfunction
11 endclass

```

Fig. 2: The **addr_txn** class

The next three classes implement some policies for the **addr_txn** class. A list of valid address ranges can be stored in the **ranges** array. The **addr_permit_policy** will choose one of the ranges at random and constrain the address to be within the range, while the **addr_prohibit_policy** will exclude the address from randomizing inside any of the ranges in its list.

```

1  class addr_policy_base extends policy_base#(addr_txn);
2      addr_range ranges[$];
3
4      function void add(addr_t min, addr_t max);
5          addr_range rng = new(min, max);
6          ranges.push_back(rng);
7      endfunction
8  endclass
9
10 class addr_permit_policy extends addr_policy_base;
11     rand int selection;
12
13     constraint c_addr_permit {
14         selection inside {[0:ranges.size()-1]};
15
16         foreach(ranges[i]) {
17             if(selection == i) {
18                 item.addr inside {[ranges[i].min:ranges[i].max - item.size]};
19             }
20         }
21     }
22 endclass
23
24 class addr_prohibit_policy extends addr_policy_base;
25     constraint c_addr_prohibit {
26         foreach(ranges[i]) {
27             !(item.addr inside {[ranges[i].min:ranges[i].max - item.size + 1]});
28         }
29     }
30 endclass

```

```

29     }
30 endclass

```

Fig. 3: Policies for constraining the `addr_txn` class

The final class shows how policies might be used. The `addr_constrained_txn` class extends `addr_txn` and defines two policies, one that permits the address to be within one of two ranges, and one that prohibits the address from being within a third range. The `addr_constrained_txn` class then passes the local `pcy` list to the parent `policy` queue.

```

1  class addr_constrained_txn extends addr_txn;
2      function new;
3          addr_permit_policy    permit = new;
4          addr_prohibit_policy  prohibit = new;
5          policy_list#(addr_txn) pcy = new;
6
7          permit.add('h00000000, 'h0000FFFF);
8          permit.add('h10000000, 'h1FFFFFFF);
9          pcy.add(permit);
10
11         prohibit.add('h13000000, 'h130FFFFF);
12         pcy.add(prohibit);
13
14         this.policy = {pcy};
15     endfunction
16 endclass

```

Fig. 4: The `addr_constrained_txn` subclass

At this point, an instance of `addr_constrained_txn` can be created and randomized like normal, and the address will be constrained based on the embedded policies.

```

1  addr_constrained_txn txn = new();
2  txn.randomize();

```

Fig. 5: Randomizing an instance of `addr_constrained_txn`

Further work on policy-based constraints has been presented since the original DVCon presentation in 2015. Kevin Vasconcellos and Jeff McNeal applied the concept to test configuration and added many nice utilities to the base policy class [3]. Chuck McClish extended the concept to manage real number values for User Defined Nettypes (UDN) and Unified Power Format (UPF) pins in an analog model [4]. Additionally, McClish defined a policy builder class that was used to generically build multiple types of policies while reducing repeated code that was shared between each policy class in the original implementation.

Although there has been extensive research on policies, this paper aims to address and provide solutions for three issues that have not been adequately resolved in previous implementations. Furthermore, a fourth problem that arose during testing of the upgraded policy package implementation will also be discussed and resolved.

II. PROBLEM #1: PARAMETERIZED POLICIES

The first problem with the above policy implementation is that because `policy_base` is parameterized to the class it constrains, different specializations cannot be grouped and indexed. The awkward consequences of this limitation become apparent when using policies with a class hierarchy. If you extend a class and add a new random field then you need a new policy type to constrain that field. The new policy type requires its own policy list, and the new list must be traversed and mapped back to the container during `pre_randomize()`.

Imagine a complex class hierarchy with several layers of inheritance and extension, and constrainable attributes on each layer (one common example is a multi-layered sequence API library, such as the example presented by Jeff Vance [5]). Using

policies becomes cumbersome in this case because each class layer requires a unique family of constraints organized into a distinct list. This stratification of policies places a burden on users to know which layer of the class hierarchy defines each attribute they want to constrain, and the name of the associated policy list for the matching policy type. For example, extending the `addr_txn` class to create a version with parity checking results in a class hierarchy that looks like this:

```

1  class addr_txn;
2      // ... unchanged from previous example
3  endclass
4
5  class addr_p_txn extends addr_txn;
6      rand bit parity;
7      rand bit parity_err;
8      rand policy_base#(addr_p_txn) addr_p_policy[$];
9
10     constraint c_parity_err {
11         soft (parity_err == 0);
12         (parity_err) ^ ($countones({addr, parity}) == 1);
13     }
14
15     function void pre_randomize;
16         super.pre_randomize();
17         foreach(addr_p_policy[i]) addr_p_policy[i].set_item(this);
18     endfunction
19 endclass
20
21 class addr_constrained_txn extends addr_p_txn;
22     function new;
23         addr_permit_policy      permit_p = new;
24         addr_prohibit_policy    prohibit_p = new;
25
26         // definition to follow in a later example - constrains parity_err bit
27         addr_parity_err_policy  parity_err_p = new;
28
29         policy_list#(addr_txn)  addr_pcy_lst = new;
30         policy_list#(addr_p_txn) addr_p_pcy_lst = new;
31
32         permit_p.add('h00000000, 'h0000FFFF);
33         permit_p.add('h10000000, 'h1FFFFFFF);
34         addr_pcy_lst.add(permit_p);
35
36         prohibit_p.add('h13000000, 'h130FFFFF);
37         addr_pcy_lst.add(prohibit_p);
38
39         parity_err_p.set(1'b1);
40         addr_p_pcy_lst.add(parity_err_p);
41
42         this.addr_policy = {addr_pcy_lst};
43         this.addr_p_policy = {addr_p_pcy_lst};
44     endfunction
45 endclass

```

Fig. 6: Modified address transaction classes with a parity checking subclass and policies included

Scaling this implementation results in a lot of repeated boilerplate code and is not very intuitive to use. Each additional subclass in a hierarchy only increases the chaos and complexity of implementing and using policies effectively.

The solution to this problem is to replace the parameterized policy base with a non-parameterized base and a parameterized extension. We chose an interface class as our non-parameterized base for the flexibility it offers over a virtual base class—specifically, our policies are bound only to implement the interface functions and not to extend a specific class implementation.

```

1  interface class policy;
2      pure virtual function void set_item(uvm_object item);
3  endclass
4
5  virtual class policy_imp#(type ITEM=uvm_object) implements policy;
6      protected rand ITEM m_item;
7
8      virtual function void set_item(uvm_object item);
9          if (!$cast(m_item, item)) begin
10              `uvm_warning("policy::set_item()", "Item/policy type mismatch")
11              this.m_item = null;
12              this.m_item.rand_mode(0);
13          end
14      endfunction: set_item
15  endclass: policy_imp
16
17  typedef policy policy_queue[$];

```

Fig. 7: The **policy** interface class and **policy_imp** class

A non-parameterized base enables all policies targeting a particular class hierarchy to be stored within a single common **policy_queue**. A parameterized template, **policy_imp**, implements the base interface and core functionality required by all policies.

One consequence of eliminating the parameter from our base type is that the policy-enabled container object, **item**, and its assignment function, **set_item()**, are no longer strongly typed. Here we make a small concession, using **uvm_object** as our default policy-enabled type. This means that all classes that implement our policies must derive from **uvm_object**, and we need to use dynamic casting to ensure that policies and their containers are type-compatible. In the example above, **item** is set to **null** and randomization is disabled when the cast fails, preventing runtime problems in the event that incompatible policies are applied.

Not much changes when it comes to defining policies; the address policies now extend **policy_imp** instead of **policy_base**, and the underlying constraints are written as implications so that they will not apply when **item** is missing.

```

1  class addr_policy extends policy_imp#(addr_txn);
2      // ... unchanged from previous example, with updated class extension
3  endclass
4
5  class addr_parity_err_policy extends policy_imp#(addr_p_txn);
6      protected bit parity_err;
7
8      constraint c_fixed_value {m_item != null -> m_item.parity_err == parity_err;}
9
10     function new(bit parity_err);
11         this.parity_err = parity_err;
12     endfunction
13 endclass
14
15 class addr_permit_policy extends addr_policy;
16     // same as before
17 endclass

```

```

18
19 class addr_prohibit_policy extends addr_policy;
20     // same as before
21 endclass

```

Fig. 8: Address policies updated to use the new policy implementation

However, the address transaction classes are simplified considerably. Only a single policy queue is required in the class hierarchy, and the vast majority of the boilerplate code has been eliminated, including all of the specialized lists. The **addr_txn** class now extends **uvm_object** to provide compatibility with the policy interface.

```

1 class addr_txn extends uvm_object;
2     rand policy_queue policies;
3     // policy is replaced with the above. All other members, constraints,
4     // and pre_randomize are unchanged from the previous example
5 endclass
6
7 class addr_p_txn extends addr_txn;
8     rand bit parity;
9     rand bit parity_err;
10    constraint c_parity_err {/*....*/}
11    // The local addr_p_policy and pre_randomize are removed. Everything
12    // else is unchanged from the previous example
13 endclass
14
15 class addr_constrained_txn extends addr_p_txn;
16     function new;
17         addr_permit_policy    permit_p    = new();
18         addr_prohibit_policy   prohibit_p  = new();
19         addr_parity_err_policy parity_err_p;
20
21         // only a single policy queue is necessary now
22         permit_p.add('h00000000, 'h0000FFFF);
23         permit_p.add('h10000000, 'h1FFFFFFF);
24         this.policies.push_back(permit_p);
25
26         prohibit_p.add('h13000000, 'h130FFFFF);
27         this.policies.push_back(prohibit_p);
28
29         parity_err_p = new(1'b1);
30         this.policies.push_back(parity_err_p);
31     endfunction
32 endclass

```

Fig. 9: Address transaction classes updated to use the new policy implementation

III. PROBLEM #2: DEFINITION LOCATION

The second problem with policies is “where do I define my policy classes?” This is not a complicated problem to solve; most users will likely wish to place their policy classes in a file or files close to the class they are constraining. However, directly embedding policy definitions within the class they constrain offers a myriad of benefits. Not only does this convention eliminate all guesswork about where to define and discover policies, but embedded policies also gain access to all members of their container class, including protected properties and methods! This privileged access enables policies to constrain attributes of a class that are not otherwise exposed, improving encapsulation.

To further optimize the organization of potentially large families of policies, we establish a convention of defining all policy classes within an embedded wrapper class called **POLICIES**. Each layer of a class hierarchy that implements policies will have its own embedded **POLICIES** wrapper, and individual **POLICIES** wrappers extend other wrappers in a manner parallel to their container classes. This parallel inheritance pattern is shown below, with **addr_p_txn::POLICIES** extending **addr_txn::POLICIES**.

```

1  class addr_txn extends uvm_object;
2      // class members, constraints, and pre_randomize unchanged from previous
3
4      class POLICIES;
5          class addr_policy extends policy_imp#(addr_txn);
6              // ... unchanged from previous standalone class example
7          endclass
8
9          class addr_permit_policy extends addr_policy;
10             // ... unchanged from previous standalone class example
11          endclass
12
13          class addr_prohibit_p_policy extends addr_policy;
14             // ... unchanged from previous standalone class example
15          endclass
16      endclass: POLICIES
17  endclass
18
19  class addr_p_txn extends addr_txn;
20      protected rand bit parity_err;
21      // other class members and constraints unchanged from previous example
22
23      class POLICIES extends addr_txn::POLICIES;
24          class addr_parity_err_policy extends policy_imp#(addr_p_txn);
25              // ... unchanged from previous example
26          endclass
27
28          static function addr_parity_err_policy PARITY_ERR(bit value);
29              PARITY_ERR = new(value);
30          endfunction
31      endclass: POLICIES
32  endclass
33
34  class addr_constrained_txn extends addr_p_txn;
35      function new;
36          addr_constrained_txn::POLICIES::addr_permit_policy permit_p = new();
37          addr_constrained_txn::POLICIES::addr_prohibit_policy prohibit_p = new();
38
39          // policy constraint value setup unchanged from previous example
40
41          this.policies.push_back(
42              addr_constrained_txn::POLICIES::PARITY_ERR(1'b1)
43          );
44      endfunction
45  endclass

```

Fig. 10: Address transaction classes with embedded policies

This example also shows how we can define static constructor functions within **POLICIES** wrappers. This practice further

reduces the cost of using policies since we can instantiate and initialize them with a single call, as demonstrated with the call to `addr_constrained_txn::POLICIES::PARITY_ERR()` . Note that although the `PARITY_ERR` constructor is defined in `addr_p_txn::POLICIES` , it is accessible through `addr_constrained_txn::POLICIES` because of the wrapper class inheritance. The `POLICIES::` scoping layer even helps to make code more readable and easy to understand.

What's more, the `parity_err` property has now been defined as `protected` , preventing anything but our `PARITY_ERR` policy from manipulating that "knob." In fact, a more advanced use of policies might define all members of a target class as `protected`, restricting the setting of fields exclusively through policies and reading through accessor functions, thus encouraging maximum encapsulation/loose coupling, which reduces the cost of maintaining and enhancing code and prevents bugs from cascading into classes that use policy-enabled classes.

IV. PROBLEM #3: BOILERPLATE OVERLOAD

The third problem with using policies is that policies are relatively expensive to define since you need at a minimum: a class definition, a constructor, and a constraint. This will be relatively unavoidable for complex policies, such as those defining a relationship between multiple specific class attributes. For generic policies, such as equality constraints (property equals X), range constraints (property between Y and Z), or set membership (keyword `inside`) constraints, macros can be used to drastically reduce the expense and risk of defining common policies. The macros are responsible for creating the specialized policy class for the required constraint within the target class, as well as a static constructor function that is used to create new policy instances of the class. Additional macros are utilized for setting up the embedded `POLICIES` class within the target class. These macros can be used hand in hand with the non-macro policy classes needed for complex constraints, if necessary.

```

1  // Fixed-value policy class and constructor macro
2  `define fixed_policy(POLICY, FIELD, TYPE)                                \
3  `m_fixed_policy_class(POLICY, FIELD, TYPE)                              \
4  `m_fixed_policy_constructor(POLICY, TYPE)                                \
5
6  `define m_fixed_policy_class(POLICY, FIELD, TYPE)                        \
7      class POLICY`_policy extends base_policy;                          \
8          protected TYPE l_val;                                           \
9
10         constraint c_fixed_value {                                       \
11             (m_item != null) -> (m_item.FIELD == TYPE'(l_val));        \
12         }                                                                \
13
14         function new(TYPE value);                                        \
15             this.l_val = value;                                           \
16         endfunction                                                      \
17     endclass: POLICY`_policy
18
19 `define m_fixed_policy_constructor(POLICY, TYPE)                          \
20     static function POLICY`_policy POLICY(TYPE value);                  \
21         POLICY = new(value);                                              \
22     endfunction: POLICY

```

Fig. 11: Macros for setting up the embedded `POLICIES` class and a fixed value policy

This example includes a ``fixed_policy` macro, which wraps two additional macros responsible for creating a policy class and a static constructor for the class. This ``fixed_policy` example policy class lets you constrain a property to a fixed value. A more complete macro definition can be found in Appendix B. The appendix includes ``start_policies` , ``start_extended_policies` , and ``end_policies` macros that are used to create the embedded `POLICIES` class within the constrained class instead of using hard-coded `class` and `endclass` statements. They set up class inheritance as needed and create a local typedef for the `policy_imp` parameterized type.

```

1  class addr_txn extends uvm_object;

```



```

2      // class members, constraints, and pre_randomize unchanged from previous
  ↪  example
3
4      `start_policies(addr_txn)
5          `include "addr_policies.svh"
6      `end_policies
7  endclass
8
9  class addr_p_txn extends addr_txn;
10     // class members and constraints unchanged from previous example
11
12     `start_extended_policies(addr_p_txn, addr_txn)
13         `fixed_policy(PARITY_ERR, parity_err, bit)
14     `end_policies
15 endclass
16
17 class addr_constrained_txn extends addr_p_txn;
18     // ... unchanged from previous example
19 endclass

```

Fig. 12: Simplified address transaction classes using the policy macros

The base `addr_txn` class has complex policies with a relationship between the `addr` and `size` fields, so rather than creating a policy macro that will only be used once, they can either be left as-is within the embedded policies class, or moved to a separate file and included with ``include` as was done here to keep the transaction class simple. The child parity transaction class is able to use the ``fixed_policy` macro to constrain the `parity_err` field. The constraint block remains the same as the previous example.

V. PROBLEM #4: UNEXPECTED POLICY REUSE BEHAVIOR AND OPTIMIZING FOR LIGHTWEIGHT POLICIES

A fourth problem we noticed in our initial usage of policy classes was occasional unexpected behavior when attempting to reuse a policy. It is hard to quantify and we never spent time to create a reproducible test case, but some of the policies behaved like they “remembered” being randomized when used repeatedly and wouldn’t reapply their constraints during subsequent `randomize` calls. Rather than spending time to diagnose and fix this issue, we were focusing on transitioning our policy approach to a “use once and discard” approach, where we would treat a policy as disposable and apply fresh policy instances before re-randomizing a target object. This improvement happened to resolve the randomization issue as well, so we never had a reason to circle back.

Fortunately, creating new policy instances for each randomization is not prohibitively expensive. We anticipated that we would need hundreds or thousands of policies per test, so we aimed to minimize the memory requirements and streamline functionality as much as possible. To minimize the footprint of policies, we chose to implement them as simple classes rather than extensions of `uvm_object` (the policy container itself still inherits from `uvm_object`). We duplicated some useful capabilities such as a basic copy routine, while omitting more expensive functionality such as factory compatibility or object comparison. The buggy randomization behavior implied we needed to pass in fresh policy instances each time we randomized an object. Two features helped to accomplish this: static constructor functions and a `copy` method.

```

1  static function addr_parity_err_policy PARITY_ERR(bit value);
2      PARITY_ERR = new(value);
3  endfunction
4  ...
5  this.policies.push_back(addr_constrained_txn::POLICIES::PARITY_ERR(1'b1));

```

Fig. 13: Example static constructor function from the `PARITY_ERR` policy class

Using static functions to create new instances of the policies provides an easy way to create and use a policy with a single function call. The policy is created and immediately pushed onto the policy queue, then discarded and recreated the next time it is needed.

```

1 interface class policy;
2     ...
3     pure virtual function policy copy();
4 endclass: policy

```

Fig. 14: The **copy** method in the **policy** interface class

Adding a **copy** method to the policy classes provides a way for a policy to return a fresh copy of itself with the same configuration. This is useful when a policy needs to be reused multiple times. The policy can be copied and pushed onto the policy queue multiple times, or onto other policy queues, and each copy will constrain the field separately. The **copy** method is **pure virtual**, so it must be implemented by each policy class.

After migrating to this pattern of reuse, the issues we were seeing with unapplied constraints disappeared.

VI. MORE IMPROVEMENTS TO THE POLICY PACKAGE

The examples presented so far are functional, but are lacking many features that would be useful in a real-world implementation. The following examples will present additional improvements to the policy package that will make it more practical and efficient to use.

A. Expanding the **policy** interface class

The following **policy** interface class adds additional methods for managing a policy.

```

1     pure virtual function string type_name();
2     pure virtual function string description();
3     pure virtual function bit item_is_compatible(uvm_object item);
4     pure virtual function void set_item(uvm_object item);
5     pure virtual function policy copy();
6
7 endclass: policy

```

Fig. 15: Expanded **policy** interface class

The **name**, **description**, and **copy** methods are implemented by the policy (or policy macro) directly and provide reporting information useful when printing messages about the policy to the log for the former two, or specific behavior for making a copy for the latter. The remaining three methods are implemented by **policy_imp** and are shared by all policies.

B. Better type safety checking and reporting in **policy_imp** methods

Some of the benefits of above methods can be seen by examining the new **set_item** method used by **policy_imp**.

```

1 virtual function void set_item(uvm_object item);
2     if (item == null) begin
3         `uvm_error("policy::set_item()", "NULL item passed")
4     end else if ((this.item_is_compatible(item)) && $cast(this.m_item, item)) begin
5         `uvm_info(
6             "policy::set_item()",
7             $sformatf(
8                 "policy <%s> applied to item <%s>: %s",
9                 this.name(), item.get_name(), this.description()
10            ),
11            UVM_FULL
12        )
13        this.m_item.rand_mode( 1 );
14
15

```

```

16     end else begin
17         `uvm_warning(
18             "policy::set_item()",
19             $sformatf(
20                 "Item <%s> type <%s> is not compatible with policy <%s> type <%s>",
21                 item.get_name(), item.get_type_name(), this.name(), this.type_name()
22             )
23         )
24         this.m_item = null;
25         this.m_item.rand_mode( 0 );
26     end
27 endfunction: set_item

```

Fig. 16: **set_item** method from **policy_imp**

The **set_item** method makes use of all the reporting methods to provide detailed log messages when **set_item** succeeds or fails. Additionally, the **item_is_compatible** is used before the **\\$cast** method is called and the **rand_mode** state is kept consistent with the result of the cast.

C. Replacing **policy_list** with **policy_queue**

Eagle-eyed readers might have noticed the lack of presence of a **policy_list** class in any of the examples above after migrating to the improved policy interface. Rather, a single typedef is all that is necessary to manage policies in a class.

```

1 typedef policy policy_queue[$];

```

Fig. 17: The **policy_queue** typedef

The **policy_queue** type is capable of storing any policy that implements the **policy** interface. The default queue methods are sufficient for aggregating policies, and in practice we found that using policy queues as containers was more efficient than **policy_list** instances. For example, for functions expecting a **policy_queue** argument we can directly pass in array literals populated by calls to static constructor functions, allowing us to define, initialize, aggregate, and pass policies all in a single line of code!

D. Provide a common implementation for policy-enabled classes using the **policy_container** interface and **policy_object** mixin

Our solution encourages loosely coupled code by defining APIs that pass **policy_queue** arguments and hide a target object's policies queue from other classes.

```

1     pure virtual function bit has_policies();
2
3     // Assignments
4     pure virtual function void set_policies(policy_queue policies);
5     pure virtual function void add_policies(policy_queue policies);
6     pure virtual function void clear_policies();
7
8     // Access
9     pure virtual function policy_queue get_policies();
10
11    // Copy
12    pure virtual function policy_queue copy_policies();
13
14 endclass: policy_container

```

Fig. 18: The **policy_container** interface class

This interface class should be implemented by any class that needs to use policies. The method implementations can then be used to manipulate a policy queue as needed, including adding to, replacing, or clearing the queue, and returning a handle to the queue or a copy of the queue to use elsewhere.

The **policy_container** interface class allows for a base class implementation that can be used to manage constraints across the entire class hierarchy.

```

1  class policy_object #(type BASE=uvvm_object) extends BASE implements
   ↪  policy_container;
2
3      protected policy_queue m_policies;
4
5      // Queries
6      virtual function bit has_policies();
7          // returns true/false based on size of m_policies
8      endfunction: has_policies
9
10     // Assignments
11     virtual function void set_policies(policy_queue policies);
12         // sets m_policies to a new queue of policies
13     endfunction: set_policies
14
15     virtual function void add_policies(policy_queue policies);
16         // adds new policies to m_policies
17     endfunction: add_policies
18
19     virtual function void clear_policies();
20         // clears m_policies
21     endfunction: clear_policies
22
23     // Access
24     virtual function policy_queue get_policies();
25         // return a handle to m_policies
26     endfunction: get_policies
27
28     // Copy
29     virtual function policy_queue copy_policies();
30         // a copy of m_policies
31     endfunction: copy_policies
32 endclass: policy_object

```

Fig. 19: A **policy_object** base class implementation

This base class implements all the **policy_container** functionality (a complete example is available in Appendix A). Packaging a common implementation and interface into a mixin enables sharing a common implementation and API for policy support among any classes that might benefit from the use of policies, as seen in the following example.

```

1  // Use policy_object for transactions
2  class base_txn extends policy_object #(uvvm_sequence_item);
3
4  // Use policy_object for sequences
5  class base_seq #(type REQ=uvvm_sequence_item, RSP=REQ) extends policy_object #(
   ↪  uvvm_sequence #(REQ, RSP) );

```

```

6
7 // Use policy_object for configuration objects
8 class cfg_object extends policy_object #(uvm_object);

```

Fig. 20: Example classes using the `policy_object` base class

E. Protecting the policy queue enforces loosely coupled code

A subtle but significant additional benefit to using a base `policy_object` class along with the `policy_container` API is the ability to mark the container’s policy queue as protected and prevent direct access to it.

The original implementation called `set_item` on each policy during the `pre_randomize` stage. That was necessary because the `policy_queue` was public, so there was nothing to prevent callers from adding policies without linking them to the target class.

Using an interface class and making the implementation private means the callers may only set or add policies using the available interface class routines, and because those routines are solely responsible for applying policies, they can also check compatibility (and filter incompatible policies) and call `set_item` immediately. This can be seen in the example `policy_object` implementation above, in the usage of the protected function `try_add_policy`.

By calling `set_item` when the policy is added to the queue, all policies will be associated with the target item automatically, so there is no need to do it during `pre_randomize`. This eliminates an easily-overlooked requirement for classes extending `policy_object` to make sure they call `super.pre_randomize()` in their local `pre_randomize` function.

VII. CONCLUSION

The improvements to the policy package presented in this paper provide a more robust and efficient implementation of policy-based constraints for SystemVerilog. The policy package is now capable of managing constraints across an entire class hierarchy, and the policy definitions are tightly paired with the class they constrain. The use of macros reduces the expense of defining common policies, while still allowing great flexibility in any custom policies necessary.

The complete policy package is available in Appendix A, with additional macro examples available in Appendix B. A functional package is also available for download [6] which can be included directly in a project to start using policies immediately.

REFERENCES

- [1] J. Dickol, “SystemVerilog Constraint Layering via Reusable Randomization Policy Classes,” DVCon, 2015.
- [2] J. Dickol, “Complex Constraints: Unleashing the Power of the VCS Constraint Solver,” SNUG Austin, September 29, 2016.
- [3] K. Vasconcellos, J. McNeal, “Configuration Conundrum: Managing Test Configuration with a Bite-Sized Solution,” DVCon, 2021.
- [4] C. McClish, “Bi-Directional UVM Agents and Complex Stimulus Generation for UDN and UPF Pins,” DVCon, 2021.
- [5] J. Vance, J. Montesano, M. Litterick, J. Sprott, “Be a Sequence Pro to Avoid Bad Con Sequences,” DVCon, 2019.
- [6] D. Mills, C. Haldane, “policy_pkg Source Code,” https://github.com/DillanCMills/policy_pkg.

APPENDIX A POLICY PACKAGE

A. *policy_pkg.sv*

```
1  `include "uvm_macros.svh"
2
3  package policy_pkg;
4
5      import uvm_pkg::*;
6
7      `include "policy.svh"
8
9      typedef policy policy_queue[$];
10
11     `include "policy_imp.svh"
12
13     `include "policy_container.svh"
14     `include "policy_object.svh"
15
16 endpackage: policy_pkg
```

B. *policy.svh*

```
1  interface class policy;
2
3      pure virtual function string name();
4      pure virtual function string type_name();
5      pure virtual function string description();
6      pure virtual function bit item_is_compatible(uvm_object item);
7      pure virtual function void set_item(uvm_object item);
8      pure virtual function policy copy();
9
10 endclass: policy
```

C. *policy_imp.svh*

```
1  virtual class policy_imp #(type ITEM=uvm_object) implements policy;
2
3      protected rand ITEM m_item;
4
5      pure virtual function string name();
6      pure virtual function string description();
7      pure virtual function policy copy();
8
9      virtual function string type_name();
10         return( ITEM::type_name );
11     endfunction: type_name
12
13     virtual function bit item_is_compatible(uvm_object item);
14         ITEM local_item;
15
16         return( (item != null) && ($cast(local_item, item)) );
```

```

17     endfunction: item_is_compatible
18
19     virtual function void set_item(uvm_object item);
20         if (item == null) begin
21             `uvm_error("policy::set_item()", "NULL item passed")
22
23         end else if ((this.item_is_compatible(item)) && $cast(this.m_item, item))
↪     begin
24             `uvm_info(
25                 "policy::set_item()",
26                 $sformatf(
27                     "policy <%s> applied to item <%s>: %s",
28                     this.name(), item.get_name(), this.description()
29                 ),
30                 UVM_FULL
31             )
32             this.m_item.rand_mode( 1 );
33
34         end else begin
35             `uvm_warning(
36                 "policy::set_item()",
37                 $sformatf(
38                     "Cannot apply policy '%0s' of type '%0s' to target object '%0s'
↪     of incompatible type '%0s'",
39                     this.name(), ITEM::type_name(), item.get_name(),
↪     item.get_type_name()
40                 )
41             )
42             this.m_item = null;
43             this.m_item.rand_mode( 0 );
44         end
45     endfunction: set_item
46
47 endclass: policy_imp

```

D. policy_container.svh

```

1 interface class policy_container;
2
3     // Queries
4     pure virtual function bit has_policies();
5
6     // Assignments
7     pure virtual function void set_policies(policy_queue policies);
8     pure virtual function void add_policies(policy_queue policies);
9     pure virtual function void clear_policies();
10
11     // Access
12     pure virtual function policy_queue get_policies();
13
14     // Copy
15     pure virtual function policy_queue copy_policies();

```

```

16
17 endclass: policy_container

```

E. policy_object.svh

```

1 class policy_object #(type BASE=uvmm_object) extends BASE implements
  ↪ policy_container;
2
3     protected policy_queue m_policies;
4
5     // Queries
6     virtual function bit has_policies();
7         return( this.m_policies.size() > 0 );
8     endfunction: has_policies
9
10    // Assignments
11    virtual function void set_policies(policy_queue policies);
12        if( this.has_policies() ) begin
13            `uvmm_warning( "policy", "set_policies() replacing existing policies" )
14        end
15        this.m_policies = {};
16        foreach( policies[i] ) try_add_policy( policies[i] );
17    endfunction: set_policies
18
19    virtual function void add_policies(policy_queue policies);
20        foreach( policies[i] ) try_add_policy( policies[i] );
21    endfunction: add_policies
22
23    virtual function void clear_policies();
24        if ( this.has_policies() ) begin
25            `uvmm_info( "policy", $sformatf("clearing [%0d] policies from %s",
  ↪ this.m_policies.size(), this.get_name()), UVM_FULL)
26        end
27        this.m_policies = {};
28    endfunction: clear_policies
29
30    // Access
31    virtual function policy_queue get_policies();
32        return( this.m_policies );
33    endfunction: get_policies
34
35    // Copy
36    virtual function policy_queue copy_policies();
37        copy_policies = {};
38        foreach( this.m_policies[i] ) begin
39            copy_policies.push_back( this.m_policies[i].copy() );
40        end
41    endfunction: copy_policies
42
43    protected function void try_add_policy( policy new_policy );
44        if (new_policy.item_is_compatible(this)) begin
45            `uvmm_info( "policy", $sformatf("adding policy %s to %s",
  ↪ new_policy.name, this.get_name()), UVM_FULL)
46            new_policy.set_item( this );

```



```
47         this.m_policies.push_back( new_policy );
48     end else begin
49         `uvm_warning( "policy", $sformatf("policy %s not compatible with target
↩ %s", new_policy.name, this.get_name()))
50     end
51     endfunction: try_add_policy
52 endclass: policy_object
```

APPENDIX B POLICY MACROS

A. *policy_macros.svh*

```

1  `ifndef __POLICY_MACROS__
2  `define __POLICY_MACROS__
3
4  //=====
5  // Embedded POLICIES class macros
6  //=====
7
8  `define start_policies(cls)                                \
9  class POLICIES;                                           \
10 `m_base_policy(cls)
11
12 `define start_extended_policies(cls, parent)              \
13 class POLICIES extends parent::POLICIES;                 \
14 `m_base_policy(cls)
15
16 `define end_policies                                     \
17 endclass: POLICIES
18
19 `define m_base_policy(cls)                                \
20     typedef policy_imp#cls) base_policy;
21
22
23 //=====
24 // Policy template macros
25 //=====
26
27 `include "constant_policy.svh"
28 `include "fixed_policy.svh"
29 `include "member_policy.svh"
30 `include "range_policy.svh"
31
32 `endif

```

B. *constant_policy.svh*

```

1  // Full policy definition
2  `define constant_policy(POLICY, FIELD, TYPE, CONST)      \
3  `m_const_policy_class(POLICY, FIELD, TYPE, CONST)        \
4  `m_const_policy_constructor(POLICY)
5
6  // Policy class definition
7  `define m_const_policy_class(POLICY, FIELD, TYPE, CONST) \
8      class POLICY`_policy extends base_policy            \
9
10         constraint c_policy_constraint {                  \
11             (m_item != null) -> (m_item.FIELD == TYPE'(CONST));
12         }
13
14         function new();

```

```

15     endfunction: new \
16 \
17     virtual function string name(); \
18         return (`"POLICY`"); \
19     endfunction: name \
20 \
21     virtual function string description(); \
22         return (`"(FIELD==CONST)`"); \
23     endfunction: description \
24 \
25     virtual function policy copy(); \
26         copy = new(); \
27     endfunction: copy \
28     endclass: POLICY``_policy
29
30 // Policy constructor definition
31 `define m_const_policy_constructor(POLICY) \
32     static function POLICY``_policy POLICY(); \
33     POLICY = new(); \
34     endfunction: POLICY

```

C. fixed_policy.svh

```

1 // Full policy definition
2 `define fixed_policy(POLICY, FIELD, TYPE, RADIX="%0p") \
3     `m_fixed_policy_class(POLICY, FIELD, TYPE, RADIX) \
4     `m_fixed_policy_constructor(POLICY, TYPE, RADIX)
5
6 // Policy class definition
7 `define m_fixed_policy_class(POLICY, FIELD, TYPE, RADIX="%0p") \
8     class POLICY``_policy extends base_policy \
9         local TYPE          l_val; \
10         local string        l_radix=RADIX; \
11 \
12         constraint c_policy_constraint { \
13             (m_item != null) -> (m_item.FIELD == TYPE'(l_val)); \
14         } \
15 \
16         function new(TYPE value, string radix=RADIX); \
17             this.set_value(value); \
18             this.set_radix(radix); \
19         endfunction: new \
20 \
21         virtual function string name(); \
22             return (`"POLICY`"); \
23         endfunction: name \
24 \
25         virtual function string description(); \
26             return ({ \
27                 `(FIELD==", \
28                 $sformatf(l_radix, l_val), \
29                 `")`" \
30             }); \
31         endfunction: description \

```

```

32
33     virtual function policy copy();
34         copy = new(l_val, l_radix);
35     endfunction: copy
36
37     virtual function void set_value(TYPE value);
38         this.l_val = value;
39     endfunction: set_value
40
41     virtual function TYPE get_value();
42         return (this.l_val);
43     endfunction: get_value
44
45     virtual function void set_radix(string radix);
46         this.l_radix = radix;
47     endfunction: set_radix
48
49     virtual function string get_radix();
50         return (this.l_radix);
51     endfunction: get_radix
52 endclass: POLICY``_policy
53
54 // Policy constructor definition
55 `define m_fixed_policy_constructor(POLICY, TYPE, RADIX="%0p")
56     static function POLICY``_policy POLICY(TYPE value, string radix=RADIX);
57         POLICY = new(value, radix);
58     endfunction: POLICY

```

D. member_policy.svh

```

1 // Full policy definition
2 `define member_policy(POLICY, FIELD, TYPE, RADIX="%0p")
3 `m_member_policy_class(POLICY, FIELD, TYPE, RADIX)
4 `m_member_policy_constructor(POLICY, TYPE, RADIX)
5
6 // Policy class definition
7 `define m_member_policy_class(POLICY, FIELD, TYPE, RADIX="%0p")
8     class POLICY``_policy extends base_policy
9         typedef TYPE          l_field_array_t[];
10
11         local l_field_array_t  m_values;
12         local bit              l_exclude;
13         local string            m_radix=RADIX;
14
15         constraint c_policy_constraint {
16             (m_item != null) ->
17                 ((l_exclude) ^ (m_item.FIELD inside {m_values}));
18         }
19
20         function new(
21             l_field_array_t values,
22             bit              exclude=1'b0,
23             string            radix=RADIX
24         );

```

```

25         this.set_values(values);
26         this.set_exclude(exclude);
27         this.set_radix(radix);
28     endfunction: new
29
30     virtual function string name();
31         return (`"POLICY`");
32     endfunction: name
33
34     virtual function string description();
35         string values_str = "";
36
37         foreach(m_values[i])
38             values_str = {
39                 values_str,
40                 $sformatf(m_radix, m_values[i]),
41                 i == m_values.size()-1 ? "" : ", ";
42
43         return ({
44             `"POLICY(FIELD ",
45             l_exclude ? "outside {" : "inside {" ,
46             values_str,
47             `"}}`"
48         });
49     endfunction: description
50
51     virtual function policy copy();
52         copy = new(m_values, l_exclude, m_radix);
53     endfunction: copy
54
55     virtual function void set_values(l_field_array_t values);
56         this.m_values = values;
57     endfunction: set_values
58
59     virtual function l_field_array_t get_values();
60         l_field_array_t l_array;
61         foreach (this.m_values[i])
62             l_array[i] = this.m_values[i];
63         return (l_array);
64     endfunction: get_values
65
66     virtual function void set_exclude(bit exclude);
67         this.l_exclude = exclude;
68     endfunction: set_exclude
69
70     virtual function bit get_exclude();
71         return (this.l_exclude);
72     endfunction: get_exclude
73
74     virtual function void set_radix(string radix);
75         this.m_radix = radix;
76     endfunction: set_radix
77
78     virtual function string get_radix();

```

```

79         return (this.m_radix);
80     endfunction: get_radix
81 endclass: POLICY`*_policy
82
83 // Policy constructor definition
84 `define m_member_policy_constructor(POLICY, TYPE, RADIX="%0p")
85     typedef TYPE POLICY`*_array_t[];
86     static function POLICY`*_policy POLICY(
87         POLICY`*_array_t values,
88         bit                exclude=1'b0,
89         string              radix=RADIX
90     );
91         POLICY = new(values, exclude, radix);
92     endfunction: POLICY

```

E. range_policy.svh

```

1 // Full policy definition
2 `define range_policy(POLICY, FIELD, TYPE, RADIX="%0p")
3 `m_range_policy_class(POLICY, FIELD, TYPE, RADIX)
4 `m_range_policy_constructor(POLICY, TYPE, RADIX)
5
6 // Policy class definition
7 `define m_range_policy_class(POLICY, FIELD, TYPE, RADIX="%0p")
8     class POLICY`*_policy extends base_policy
9         local TYPE        l_low;
10        local TYPE        l_high;
11        local bit         l_exclude;
12        local string       l_radix=RADIX;
13
14        constraint c_policy_constraint {
15            (m_item != null) -> (
16                (l_exclude) ^
17                ((m_item.FIELD >= l_low && m_item.FIELD <= l_high))
18            );
19        }
20
21        function new(
22            TYPE    low,
23            TYPE    high=low,
24            bit     exclude=1'b0,
25            string  radix=RADIX
26        );
27            this.set_range(low, high);
28            this.set_exclude(exclude);
29            this.set_radix(radix);
30        endfunction: new
31
32        virtual function string name();
33            return ("POLICY`");
34        endfunction: name
35
36        virtual function string description();
37            return ({

```

```

38         ~(FIELD ",
39         l_exclude ? "outside [" : "inside [",
40         $sformatf(l_radix, l_low),
41         ", ",
42         $sformatf(l_radix, l_high),
43         ~"])"
44     });
45     endfunction: description
46
47     virtual function policy copy();
48         copy = new(l_low, l_high, l_exclude, l_radix);
49     endfunction: copy
50
51     virtual function void set_range(TYPE low, TYPE high);
52         if (low <= high) begin
53             this.l_low = low;
54             this.l_high = high;
55         end else begin
56             this.l_low = high;
57             this.l_high = low;
58         end
59     endfunction: set_range
60
61     virtual function TYPE get_low();
62         return (this.l_low);
63     endfunction: get_low
64
65     virtual function TYPE get_high();
66         return (this.l_high);
67     endfunction: get_high
68
69     virtual function void set_exclude(bit exclude);
70         this.l_exclude = exclude;
71     endfunction: set_exclude
72
73     virtual function bit get_exclude();
74         return (this.l_exclude);
75     endfunction: get_exclude
76
77     virtual function void set_radix(string radix);
78         this.l_radix = radix;
79     endfunction: set_radix
80
81     virtual function string get_radix();
82         return (this.l_radix);
83     endfunction: get_radix
84     endclass: POLICY`_policy
85
86 // Policy constructor definition
87 `define m_range_policy_constructor(POLICY, TYPE, RADIX="%0p")
88     static function POLICY`_policy POLICY(
89         TYPE    low,
90         TYPE    high=low,
91         bit     exclude=1'b0,

```

```
92     string radix=RADIX  
93 );  
94     POLICY = new(low, high, exclude, radix);  
95 endfunction: POLICY
```

```
\  
\  
\
```